
 Système d'équation linéaires avec LinearAlgebra
© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

Ce doument est une révision de celui produit en 2004. L'objectif principal de cette feuille Maple est d'initier 
l'élève à la résolution de systèmes d'équations linéaires. L'extension LinearAlgebra  dispose de toutes les 
macro-commandes nécessaires à cette initiation. Dans un premier temps, l'élève apprendra à développer pas à pas 
la méthode de réduction de Gauss en échelonnant la matrice augmentée d'un système d'équations linéaires à l'aide 
des opérations élémentaires sur les lignes et à l'aide de la méthode du pivot. Dans un deuxième temps, l'élève sera 
amené à discuter la nature des solutions d'un système d'équations linéaires à l'aide des macro-commandes 
permettant d'obtenir directement la forme échelonnée ou échelonnée réduite de la matrice augmentée du système à
résoudre. Finalement, l'élève verra comment procéder à l'inversion de matrices régulières par la méthode de 
réduction de Gauss.

Bonne lecture à tous !

 Ce document Maple est exécutable avec la version 2020.2

Initialisation
restart;

with(LinearAlgebra):

Macro-commande LinearAlgebra[GenerateMatrix]
Soit le système d'équations linéaires S suivant:
                  -    +     = 
       S:        +       -     =  2
               +   -   =  1

S:=[  x[1] - 2*x[2]  +   x[3]  = -1 ,

      x[1] +   x[2]  -   x[3]  =  2 ,

    3*x[1] + 6*x[2]  - 4*x[3]  =  1 ];
    

La macro-commande GenerateMatrix de la bibilothèque LinearAlgebra permet de créer, à la fois, la 
matrice des coefficients et la matrice colonne (vecteur colonne) des termes constants d'un système d'équations 
linéaires: 

le premier argument de la macro-commande doit être la séquence des équations formant le système. 
Cette séquence qui doit être formulée comme un objet de type set ou de type list;

le deuxième argument de la macro-commande doit être la liste des inconnues du système;

le troisième argument, « augmented=true » est optionnel: cette option permet de créer directement 
la matrice augmentée du système d'équations linéaires.

Attention:
 Si on omet le troisième argument, le résultat de GenerateMatrix est une séquence de deux objets: soit 
la matrice des coefficients du système d'équations linéaires ainsi que la matrice colonne des termes 
constants.

Si on précise ce troisième argument, le résultat de GenerateMatrix est exactement la matrice augmentée
du système d'équations linéaires. Le résultat n'est donc pas une séquence d'objets.

A,b:=GenerateMatrix(S,[seq(x[i],i=1..3)]);

whattype(A);
Matrix

whattype(b);
Vectorcolumn

`[A|b]`:=GenerateMatrix(S,[seq(x[i],i=1..3)],augmented=true);

On peut très bien obtenir la matrice augmentée à partir de la matrice A et du vecteur colonne b obtenus 
précédemment. Utilisons le raccourci ligne <|>.

`[A|b]`:=<A|b>;

whattype(`[A|b]`);
Matrix

Méthode de réduction de Gauss

Opérations élémentaires de lignes avec la macro-commande LinearAlgebra
[RowOperation]
Voici la syntaxe Maple des opérations élémentaires sur les lignes d'une matrice M.
  (a) permutation de la i-ième ligne et de la j-ième ligne .........................
  (b) multiplication de la i-ième ligne par un scalaire  ...........................
  (c) addition à la i-ième ligne d'un multiple k de la j-ième ligne .............  

Effectuons l'échelonnage de la matrice augmentée [ ] du système d'équations linéaires S. 

Rappelons-nous le détail de la  matrice `[A | b]`.
'`[A|b]`'=`[A|b]`;

Additionnons à la deuxième ligne, (-1) fois la première ligne.
M1:=RowOperation(`[A|b]`,[2,1],-1);

Additionnons à la troisième ligne, (-3) fois la première ligne.
M2:=RowOperation(M1,[3,1],-3);

Additionnons à la troisième ligne, (-4) fois la deuxième ligne
M3:=RowOperation(M2,[3,2],-4);

La forme de la matrice M3 est appelée matrice échelonnée (selon les lignes):
le premier élément non nul de chaque ligne est situé à droite du premier élément non nul de la ligne 
précédente. Cet élément est appelé le pivot de la ligne.
Dans certains manuels, on exige que ce premier élément non nul soit l'unité, c'est-à-dire que chaque 
pivot doit être égale à l'unité.

toutes les lignes contenant que des zéros (s'il y a de telles lignes) sont situées sous les lignes qui 
contiennent des éléments non nuls.

En associant la matrice M3 au système S3 ci-dessous,
                       -    +     = 
           :             3    - 2   =  3
                                               =  
on résout facilement, avec un calcul manu scriptus, le système S3 par substitution à rebours. Pour ce faire, 
à partir de la troisième équation, on déduirait que la solution du système d'équations linéaires S3 est 

, ensuite que  et finalement que  .

Par l'application des opérations élémentaires sur les lignes, on a que le système S3 ~ S. Ce qui montre que 
le système d'équations linéaires original S est un système possédant une solution unique.

La macro-commande BackwardSubstitute de la bibliothèque LinearAlgebra automatise la 
substitution à rebours lorsqu'on l'applique à la matrice échelonnée du système d'équations S.

Solution=BackwardSubstitute(M3);

whattype(rhs(%));
Vectorcolumn

Le résultat un objet de type Vector dont l'ordre des composantes correspond à l'ordre des inconnues.

Formulons la réponse comme suit dans la zone de texte ci-dessous:

 Réponse: ,  et  ou E.-S.= .

Poursuivons avec les opérations élémentaires sur les lignes afin de rendre unitaire chaque pivot de la 
matrice échelonnée.

Pour déduire la forme échelonnée de la matrice A,  il suffit, ici, de multiplier la deuxième ligne de la 
matrice M3 par 1/3.

M4:=RowOperation(M3,2,1/3);

Pour obtenir le vecteur colonne solution du système, on peut, également bien sûr, appliquer la macro-
commande BackwardSubstitute sur la matrice échelonnée précédente.

Solution=BackwardSubstitute(M4);

On a formulé précédemment, dans une zone de texte, le système d'équations linéaires correspondant à la 
matrice échelonnée M3. Nous allons montré qu'il est possible aussi de le faire directement dans une zone 

de résultats.

La macro-commande GenerateEquations de la bibilothèque LinearAlgebra permet de créer un 
système d'équations linéaires à partir d'une matrice désignée comme matrice des coefficients ou comme 
matrice augmentée d'un système:

le premier argument de la macro-commande doit être la matrice considérée comme la matrice des 
coefficients du système d'équations linéaires qu'on veut créer

le deuxième argument de la macro-commande doit être la liste des inconnues du système. Si le 
deuxième argument est seulement un nom de variable (pas un objet de type list), alors c'est la 
notation indicielle correspondant à cette variable qui sera pris en compte pour les variables du 
système

si le troisième argument, qui est optionnel, est un vecteur, alors ce vecteur sera pris en compte 
comme le vecteur colonne des termes constants du système qui sera créer. Si on omet ce troisième 
argument, c'est le vecteur nul qui sera pris en compte, ce qui générera un système homogène.

Comme exemple, générons le système d'équations associé à la matrice M4. Comme cette matrice n'est pas 
la matrice des coefficients mais plutôt la matrice augmentée du système S, il faut d'abord en extraire la 
matrice des coefficients. Utilisons SubMatrix de l'extension LinearAlgebra.

Rappelons-nous d'abord la matrice M4;
M4;

A_échelon:=SubMatrix(M4,1..3,1..3);

Extrayons ensuite la quatrième colonne de la matrice M4.
b_échelon:=SubMatrix(M4,1..3,4..4);

Créons maintenant le système d'équations linéaires associé à la matrice augmentée M4
`S'`:=GenerateEquations(A_échelon,[seq(x[i],i=1..3)],b);

 Soit donc le système       

                                                



(4.1.2)(4.1.2)

> > 

> > 

> > 

(4.1.1)(4.1.1)

> > 

(3.1.13)(3.1.13)

(3.2.1)(3.2.1)

> > 

(4.1.3)(4.1.3)

                                                            

Notons que le système d'équations (le résultat de la macro-commande) est formulé en tant qu'objet de type
list.

whattype(`S'`);
list

 Macro-commande LinearAlgebra[GaussianElimination]
La macro-commande GaussianElimination permet d'obtenir automatiquement la matrice échelon (le 
pivot pas nécessairement unitaire) de la matrice augmentée du système d'équations linéaires.

G:=GaussianElimination(`[A|b]`);

 Méthode de réduction de Gauss-Jordan

 Opérations élémentaires de lignes (suite)
Les opérations élémentaires de lignes peuvent être utilisées pour transformer davantage la matrice 
échelonnée. Poursuivons les transformations de la matrice  à partir de la matrice échelonnée M3 de 
telle sorte que le pivot de chaque ligne soit l'unique élément non nul de sa colonne.

Rappelons-nous la matrice échelonnée M3
M3;

Additionnons à la deuxième ligne, 2 fois la troisième ligne.
M4:=RowOperation(M3,[2,3],2);

Additionnons à la première ligne, (-1) fois la troisième ligne.
M5:=RowOperation(M4,[1,3],-1);

Additionnons à la première ligne, 

 fois la deuxième ligne.

M6:=RowOperation(M5,[1,2],2/3);

Poursuivons les transformations afin de rendre le premier élément non nul de chaque ligne égale à l'unité.

Ici, il suffit seulement de multiplier la deuxième ligne de la matrice M6 par .

M7:=RowOperation(M6,2,1/3);

La matrice M7 satisfait la définition d'une matrice échelonnée réduite:
le pivot de chaque ligne est égal à l'unité

le pivot d'une ligne est toujours à la droite du pivot de la ligne précédente

dans la colonne du pivot de chaque ligne, il est le seul élément non nul de la colonne

s'il y a des lignes dont tous les éléments sont nuls, elles composent les dernières lignes de la matrice.

 Technique du pivotage
La matrice échelonnée réduite M7 a été obtenue en utilisant les opérations élémentaires de lignes sur la 
matrice augmentée [ ] du système d'équations linéaires S. Nous allons maintenant obtenir de nouveau la
matrice échelonnée réduite du système S mais, cette fois, par la technique du pivotage.

Rappelons-nous la matrice augmentée du système S.
M:=`[A|b]`;

La technique du pivotage consiste à 
désigner comme pivot un élément non nul  d'une matrice M.

la i-ème ligne est alors appelée ligne pivot
la j-ème colonne est alors appelée colonne pivot

additionner à chacune des autres lignes un multiple approprié de la ligne pivot afin d'annuler toutes 
les autres entrées de la colonne pivot

En désignant l'entrée  pour effectuer le pivotage, la macro-commande Pivot de la bibliothèque
LinearAlgebra annule automatiquement toutes les entrées de la j-ième colonne sauf évidemment 
l'élément  qui a été désigné comme pivot. La technique du pivotage est en quelque sorte une façon 
d'effectuer plusieurs opérations élémentaires RowOperation dans un ordre différent. 

Désignons l'élément  de la matrice M comme pivot dans la macro-commande pivot.
M1:=Pivot(M,1,1);

Désignons l'élément  comme pivot.
M2:=Pivot(M1,2,2);

Désignons l'élément  comme pivot.
M3:=Pivot(M2,3,3);

Pour obtenir la matrice échelonnée réduite du système S, il ne reste qu'à rendre unitaire l'élément  
M4:=RowOperation(M3,2,1/3);

La technique du pivotage a permis d'obtenir la matrice échelonnée réduite plus rapidement que par 
l'application seules des opérations élémentaires de lignes.

Remarque: Dans le cas où il suffit seulement de résoudre un système d'équations linéaires par la technique
du pivotage, il n'est pas nécessaire de désigner, dans l'ordre, les éléments , , , etc., comme pivot.
L'exemple ci-dessous va le montrer.

Soit le système T suivant à résoudre.
             3  - 2   - 3   = 
       T:   3  + 2   +     =  0
             2  +     +     =  1

T:=[  3*x[1] - 2*x[2] - 3*x[3] = 4 ,

      3*x[1] + 2*x[2] +   x[3] = 0 ,

      2*x[1] +   x[2] +   x[3] = 1 ];

M:=GenerateMatrix(T,[seq(x[i],i=1..3)],augmented=true);

La réalisation du pivotage peut très bien se faire dans un ordre différent.
Désignons en premier l'élément  comme pivot.

M1:=Pivot(M,3,2);

Désignons l'élément  comme pivot.
M2:=Pivot(M1,2,3);

Désignons l'élément  comme pivot.
M3:=Pivot(M2,1,1);

Multiplions la première ligne de la matrice M3 par .

M4:=RowOperation(M3,1,1/8);

Multiplions la deuxième ligne de la matrice M4 par -1.
M5:=RowOperation(M4,2,-1);

Même si le dernier résultat ne correspond pas à la forme échelonnée réduite de la matrice augmentée, on 
peut quand même déduire l'ensemble-solution du système T:  ,  et  puisque la matrice 
augmentée correspond au système équivalent ci -dessous.
                                               =    
                                               =    1
                                               =  -2

 Macro-commandes LinearAlgebra[GaussianElimination] et LinearAlgebra
[ReducedRowEchelonForm]
La macro-commande GaussianElimination de l'extension LinearAlgebra donne comme résultat une 
matrice échelonnée. Appliquons cette macro-commande à la matrice augmentée M du système T. 

M_échelonnée:=GaussianElimination(M);

Il est possible d'obtenir une matrice échelon sont introduire de fraction.
M_échelonnée:=GaussianElimination(M,'method'='FractionFree');

La macro-commande ReducedRowEchelonForm l'extension LinearAlgebra donne directement pour
résultat la forme échelonnée réduite. Appliquons cette macro-commande à la matrice augmentée M du 
système T. 

M_échelonnée_réduite:=ReducedRowEchelonForm(M);

 Nature des solutions d'un système d'équations linéaires

 Système contradictoire

Résolvons l'équation matricielle S:     =  .

Puisque le système à résoudre est présenté avec l'écriture matricielle AX = b, ce que l'on recherche est alors
une matrice colonne X satisfaisant cette équation matricielle.

Transposons en Maple l'écriture matricielle du système d'équations linéaires S.

Créons la matrice des coefficients du système S.
A:=Matrix([

    [2,3,-1],

    [1,1,2],

    [1,2,-3] ]); 

Créons la matrice colonne des inconnues du système S.
X:=Vector([x,y,z]);

C'est plus concis en créant la matrice colonne des inconnues avec le raccourci colonne <,>.
X:=<z,y,z>;

Créons la matrice colonne des termes constants du système S.
b:=<-1,12,2>;

Ainsi, on pourra donc afficher le système S avec une écriture matricielle dans une zone des résultats.
A.X = b;

Pour ensuite créer la matrice augmentée associée au système d'équations avec le racourci ligne <|>.
`A|b`:=<A|b>;

Pour déterminer la nature du système d'équations, obtenons directement la forme de Gauss-Jordan de la 
matrice A|B à l'aide de la macro-commande rref.

`A|b_échelon_réduite`:=ReducedRowEchelonForm(`A|b`);

La dernière ligne de la matrice échelonnée réduite nous montre clairement que le système à résoudre est un
système incohérent. Dans une zone de texte, reste à formuler la réponse.
Réponse: E-S = { } ( ou = :)

Par curiosité, appliquons quand même la macro-commande BackwardSubstitute sur cette matrice 
pour "voir" comment l'évaluateur traitera la requête.

X=BackwardSubstitute(`A|b_échelon_réduite`);
Error, (in LinearAlgebra:-BackwardSubstitute) inconsistent 
system

 Système cohérent avec
solution unique

Résolvons le système S:     =  .

Créons la matrice des coefficients du système S.
A:=Matrix([

   [1, -2, 3],

   [2, 2, -1],

   [1, 1, 5],

   [1, 1, -6] ]);

Créons la matrice colonne des termes constants du système S.

b:=<16,0,11,-11>;

Créons la matrice augmentée du système S.
`A|b`:=<A|b>;

Obtenons directement avec la macro-commande ReducedRowEchelonForm la forme de Gauss-
Jordan de la matrice A|B.

`A|b_échelon_réduite`:=ReducedRowEchelonForm(`A|b`);

La matrice  associée au système S nous montre que le système est un système 
cohérent avec solution unique. La matrice de Gauss-Jordan nous donne, sans calcul à faire, la solution 
du système S.

À cause de la notation matricielle du système S, la solution doit être formulée comme un vecteur 
colonne.

Réponse:  E-S  =  .

Mais, il est plus commode d'utiliser la notation transposée.
Réponse: E-S = {  }

Utilisons quand même la macro-commande BackwardSubstitute et observons le résultat qui 
s'affichera.

X=BackwardSubstitute(`A|b_échelon_réduite`);

une infinité de solutions

Résolvons le système S:     =  .

Obtenons la matrice augmentée du système S.
A:=Matrix([

    [ 0,-3,-6, 4],

    [-1,-2,-1, 3],

    [-2,-3, 0, 3],

    [ 1, 4, 5,-9]  ]):

b:=<9,1,-1,-7>:

`A|b`:=<A|b>;

Contruisons le vecteur colonne des inconnues.
X:=Vector([seq(x[i],i=1..4)]);

Obtenons directement la matrice de Gauss-Jordan de A|B
`A|b_échelon_réduite`:=ReducedRowEchelonForm(`A|b`);

La matrice A|B_échelon_réduite associée au système à résoudre nous indique que le système S possède 
une infinité de solutions. Utilisons la macro-commande backsub et observons, dans ce cas, la 
formulation du résultat.

X=BackwardSubstitute(`A|b_échelon_réduite`);

L'utilisation du trait souligné nous indique que  est une variable mathématique qui a été 
automatiquement générée par le simplificateur. Dans le contexte actuel, on doit interpréter  comme 
un paramètre. On constate donc que le système possède une infinité de solutions et que la formulation 
donnée du résultat est une formulation paramétrique où l'évaluateur a désigné la variable  comme 
étant la variable libre du système.

Dans une zone de texte alors, formulons donc l'ensemble-solution avec la notation habituelle d'un 
l'ensemble-solution dans le cas d'un système indéterminé.
Réponse: E-S =  {  | s

intéressant de reformuler, dans une zone de résultats, Solution avec une notation mathématique plus 
habituelle. Il suffira, dans ce cas, de faire une substitution de noms.

subs(_t10[1]=t,(5.2.2.4));

 Macro-commande LinearAlgebra[LinearSolve]

Résolvons de nouveau le système S:     =  mais, cette fois, en utilisant la 

macro-commande LinearSolve de la bibliothèque LinearAlgebra.
Les deux premiers arguments obligatoires de la macro-commande linsolve sont, dans l'ordre, 

soit la matrice des coefficients et la matrice colonne des termes constants 

soit la matrices des coefficients et le vecteur (colonne) des termes constants.
 
Créons de nouveau la matrice A des coefficients et le vecteur colonne b des termes constants.

A:=<< 0,-1,-2, 1>|

    <-3,-2,-3, 4>|     

    <-6,-1, 0, 5>|

    < 4, 3, 3,-9>>;

b:=<9,1,-1,-7>;

 

Résolvons le système avec la macro-commande LinearSolve en précisant l'option free='t' afin de 
désigner le symbole à utiliser pour la ou les variables libres .

X=LinearSolve(A,b,free='t');

 Inversion de matrices par la méthode de réduction de Gauss 

Trouvons la matrice inverse, si elle existe, de la matrice   en utilisant la méthode de 

réduction de Gauss.

Comme on l'a vu en classe, commençons d'abord par augmenter la matrice A avec la matrice identité.

Créons la matrice A.
A:=< <3,2,-1>|

     <2,-1,1>|

     <4,2,2> >;

`A|I`:=<A|IdentityMatrix(3)>;

(5.2.2.4)(5.2.2.4)

> > 

Reste maintenant à obtenir directement la forme de Gauss-Jordan de la matrice A|Id avec la macro-commande
ReducedRowEchelonForm.

B:=ReducedRowEchelonForm(`A|I`);

Le résultat obtenu montre clairement que la matrice A est régulière puisque la matrice B est de la forme I |
.

Pour formuler la réponse dans une zone de résultats, utilisons la macro-commande SubMatrix pour extraire la
matrice inverse de la matrice B.

A_inverse:=SubMatrix(B,1..3,4..6);

Vérification:
'A.A_inverse' = A.A_inverse;

'A_inverse.A' = A_inverse.A;


